
【注意】

　このサイト「モデル記述言語CASTによるシミュレーション」[注1]で公開されているsimcast.zip（＋extProlog）をインストールすると，モデル理論アプローチによるシミュレーション開発実行環境Simcastができあがり，本文中のモデルを実行することができる．ただし，simcastのバージョンは0908xx以降でなければならない．開発実行環境Simcastの操作方法などは，同サイトや教科書[注2]を参照されたい．
１ はじめに

　本稿では，オートマトンから問題解決システム（Solver）を呼び出して，値の授受を行ないながら利用する方法を解説する．すなわち，Solverは，外部から各種の設定値を変更されうるし，問題解決の実行結果を外部に返すことができる．そのようなSolverをここでは「関数化したSolver」と称している（図1）．また，関数化したSolverを外部から利用することを「Solverを呼び出す（callする）」ということにする．

図１　関数化した問題解決システム

　オートマトンからSolverを呼び出すことができれば，「Solverの定義ファイルを変更することなく」異なる設定値に対する解を次々と求めることが可能になるので，各変数と解の間の関係を知るのに便利である．また，関数化したSolverを状態遷移関数とするオートマトンを作成することができる．複数のオートマトンを結合させることができるので，複雑なシステムシミュレーションを行なうことが可能となる．
2 利用の概要と書式

　次の手順に従えば，Solverを関数化して利用することができる．

（1）まず，入力出力値を授受できる問題解決ユーザモデル（.setファイル）を作成しコンパイルすることにより，関数化したSolver（.pファイル）を作成しておく．

【書式】

1）solverInput.g(x)

2）solverOutput.g:=y

【意味】
1）外部からの入力をローカル変数xに代入する．

2）外部への出力（yの値）を規定のグローバル変数solverOutput.gに代入する．

（2）オートマトンから Solverを呼び出して利用する．

【書式】y:=call_solver("ファイル名", 入力)

【例】　y:=call_solver("opt1.p", x)

【意味】関数化されたシステムopt11.pにxの値を入力し，opt1.pを実行する．結果をyの値に取得する[注3]．

　これらの述語や関数の利用に関する詳細は以下で述べていく．

3 Solverの関数化（1変数）

　例として極値を求めるSolver（付録１）を取り上げる．これは，関数f(x) = x*(x - r.g)の変数xを変動させたときに極大を与えるxを求める問題解決システムである．問題解決システムの設計法については文献を参照されたい[注2]．問題解決活動の最中では，定数r.gの値は固定である．ここでは，定数r.gを外部（別ファイルに定義したオートマトン）から自由に設定変更できるようにしたい．定数r.gの値が変われば，最終状態（関数fの極大を与えるx）も変わるので，これを出力として取得するつもりである．Simcast（バージョン0908xx以降）では，外部からの入力と外部への出力は，図2のように，規定のグローバル変数solverInput.gとsolverOutput.gを媒介して授受する．


preprocess() <-> (solverInput.g(v)) -> (r.g:=v) otherwise (r.g:=4);

  ---

st(x) <-> finalstate(x), solverOutput.g:=x;
　　　　図２　入力出力値を授受するSolver（付録１の抜粋）

　外部入力の受け取りに関しては，図2のように条件つきで定義する．

　　　(solverInput.g(v)) -> (r.g:=v) otherwise (r.g:=4)
これは作法である．solverInput.g(v)は，「もしも外部入力があれば，その値を変数vに代入し，真理値が真となる」述語である．もしも外部入力がなければsolverInput.g(v)は偽となる．したがって，事前準備preprocess()の行は「もしも外部入力があるならば，定数r.gの値を外部で設定されている値とせよ．しかし外部入力がないならば，r.g:=4とせよ．」と理解することができる．ただし，条件判定部分にv:=solverInput.gやsolverInput.g(r.g)と記述してもエラーとなるので注意すること．

　また，もしも事前準備preprocess()を使用したくない場合には，初期値設定の副作用として外部入力を取得することもできる．例えば，

　　　initialstate()=x0 <-> x0:=0, e.g:=0.01, (solverInput.g(v)) -> (r.g:=v) otherwise (r.g:=4);

とすれば良い．

　このように，記述する場所がどこであれ，条件つきで定義する利点は，外部入力が無い時でもSolver単体でモデル作成，実行，修正が可能だからである． 
　一方，外部への出力に関しては，停止条件の「副作用」として定義する．図２のモデルでは，停止条件st(x)の行で，グローバル変数solverOut.gに最終状態xの値を代入している．
　このように入力出力値を授受できる問題解決ユーザモデル（.setファイル）を作成し，コンパイルすると，関数化した問題解決システム（.pファイル）を作ることができる．重要なことは，関数化した問題解決システム（今の場合は，opt11.p）を必ず作成し，削除しないことである．

　図3は，opt11.setを単体で動作確認したときのgoal(x)=-f(x)の推移を表わすグラフである．原点から関数f(x)を下って極小に達している様子が分かる．

　　　　　　　　　　　　　　図3　問題解決の実行

4 関数化したSolverの呼び出し

　次に，別ファイルに定義したオートマトンから，Solverを呼び出して実行させたい．図4は上記の関数化したSolver "opt11.p"を子プロセスとして実行させるオートマトンモデルである．

 

//opt12.set

inputsequence0()=Rs <-> Rs:=[1,2,3,4];

initialstate0()=c <-> c:=0;

delta0(c,r)=cc <-> cc:=call_solver("opt11.p",r);
　　　　図4　Solverを呼び出すオートマトンモデル

　述語cc:=call_solver("opt11.p", r) で，変数rの値をSolverに渡して実行させている．返ってきた実行結果（解）は変数ccに代入される．以上のように定義しておけば，Solverを関数として扱うことができる．

　　　　　　　　　図5　オートマトンopt12.setの実行結果

　図4を実行すると上図5のように表示される．右側は通常Solver実行中に出現するグラフウインドウであり，関数goal(x) = -f(x)の値の変化が表示されている．4回の入力がある（r=1,2,3,4)ので，4回このウインドウは書き換えられる．

　左側はSolverとオートマトンの実行処理中の入力と出力と状態の変化の様子が表示されている．最終的な入力r = 4のときの極値を与えるxは，本来は2.0であるはずだが，このウインドウでは，C=1.999998と表示されていることに注意する．その理由は，付録1のSolverが誤差0.01以内で停止するように定義されているからである．以上により，「Solverの定義ファイルを変更することなく」，異なる設定値に対する解を次々と求めることが可能になった．

　しかしながら，図5のままでは直感的に理解することが困難である．そこで文献 [注4]を参考に，グラフに点(r,cc)をプロットするように変更したものを付録2に収録した．付録2のユーザモデル"opt13.set"をコンパイルし実行した結果を図6に示す．問題解決システムの各変数と解の間の関係を知るのに便利である．

図6　入力出力値のグラフ表示

　さて，処理時間を見てみたい．ここでオートマトンへの入力系列Rs（つまりSolverへの外部入力）を，inputsequence0()=Rs <-> Rs:=[10,11,12,13]; に変更し実行すれば，極端に遅いことが分かる．その理由は，問題解決のサイクル

　　　　　　　r → r.g → x0=0 → 問題解決 → 解x → cc

毎に，いつもx0 = 0から問題解決活動を開始するからである．しかしながら，r.gとr.g’が近ければ解xとx’も近いので，「次のサイクルの開始点x0 = 前回の結果x」のほうが問題解決の時間が少なくてすむはずである．したがって「前回の解xを今回の初期状態x0として再利用」しながら問題解決活動を連続して実行できれば処理時間が早くなるはずである．次節でこれを扱う．

5　Solverのオートマトン化（2変数）

　本節では，2変数の値をSolverに渡すことを考える．例として，図7のように，オートマトンの状態遷移関数δそのものをSolverにしてみる．これにより，「前回の解を記憶しておき今回の初期状態として再利用し」ながら問題解決活動を連続して実行することが可能である．





　　　　　　　　　　図7　状態遷移関数がSolverであるオートマトン

　付録3に2変数の外部入力を取得するSolverの例（opt21.set)を収録した．opt21.setは，外部入力があれば初期状態x0と定数r.gを設定変更でき，結果（関数fの極値を与えるx）を外部に出力するモデルである．また付録4はそれを状態遷移関数とする状態機械（出力のないオートマトン）のモデルである．オートマトンからの入力(c,r)の値はSolverの(x0, r.g) に渡され，問題解決の結果（関数fの極値を与えるx）は，オートマトンの変数ccに返される．

　付録3, 4の方法の本質的な特徴は，

オートマトンの状態cの値をSolverの初期状態x0に代入し，

Solverの解xの値をオートマトンの次の状態ccに代入する

ことにある．要するにSolverの解を使い回すのである．何度も繰り返す問題解決では，まず解を得た時点で立ち止まり，設定を変えて，再びその時点から問題解決活動を行なうことになろう．この方法は，そのような連続した繰り返し問題解決のモデルとなるので，この方法を「Solverのオートマトン化」と呼んでいる．

　さて，関数がf(x) = x*(r.g -x)であるので，定数r.gが決まれば，どんな初期状態x0から出発しても同じ極値を与えるxが得られる．したがって，付録3, 4をコンパイルし実行させても，前節の図3や図4と同じ計算結果を得るのは当然である．しかしながら，入力系列を inputsequence()=Rs <-> Rs:=[10,11,12,13]; に変更し実行すれば，本節のopt22.setの実行時間は前節のopt12.setより処理時間が短くなっていることが分かる（一回目の状態遷移を除く）．

6 注意点

1）Solverのコンパイル

　先にSolverをコンパイルして実行しておくことが必要である．例題の場合は，問題解決システムopt11.pが存在しなければならない．その後，オートマトンopt12.setをコンパイルし実行する．

2）ユーザ定義の述語や関数の衝突

　Solverを他のファイルから呼び出すときの留意点は，親プロセスで定義された述語や関数が子プロセスに引き継がれることである．したがって，オートマトンで定義された「ユーザ定義の述語や関数」は，Solverで定義された「ユーザ定義の述語や関数」と衝突しないように，異なる名称をつける必要がある．

3）トレースモード

　オートマトンの実行時とSolverの実行時に，トレースモードにするか否かを聞かれるので，nを入力すること．システム開発時には役立つが，シミュレーション開発の後半や，他人に見せるとき（プレゼン用）には煩わしい．不要と感じる場合には，stdAutomatonEngine.pとstdPDsolver122.pの中の「trace mode?」関連部分をコメントアウトせよ．

4）SolverからSolverを呼び出して実行することはできない．

　もしそうしたければ，二つのSolverを第5節の方法でオートマトン化し，オートマトンとして結合させること．

7 おわりに

　古いバージョンのSimcastでは，述語initialstate()がオートマトンとSolverで衝突することを回避するためにゼロ(0)をつけていたが，新しいSimcast（バージョン0908以降）ではゼロが不要となった．また，使いやすい関数call_solver("opt11.p",r)を新規に導入した．

　その結果，すべてのSolverを簡単にオートマトンに埋め込み，オートマトンとして扱うことができるようになった．すでに確立した方法[注5]により，それらを組合わせた複雑なシステムを構築することができるだろう．それが利点である．応用例として「価格調整システム」を公開しているので参照されたい[注6]．

注

[1] 旭貴朗「モデル記述言語CASTによるシミュレーション」
　http://www2.toyo.ac.jp/~asahi/research/simulation/（2008.02.23）

[2] 高原康彦ほか『形式手法 モデル理論アプローチ：情報システム開発の基礎』日科技連(2007)  関連サイトhttp://www.geocities.jp/kisoten/（2008.04.11）

[3] 関数call_solverは開発環境Simcastに新たに追加した組込みの関数である．その内容はファイルstdAutomatonEngine.pに次のように定義した．単純にグローバル変数で入力出力の受け渡しを行なっているにすぎない．

  call_solver(SolverFname,X,Y):-


SolverInput.g:=X,


load_go(SolverFname),


solverOutput.g(Y);
[4] 技術資料「グラフによるデータ表示」技術資料（所在は[1]）

[5] 技術資料「Moore型オートマトンのネットワークモデル」（所在は[1]）

[6] 技術資料「価格調整システム」（所在は[1]）

付録

付録1　値を授受する（関数fの極値を求める）Solver

//opt11.set
func([f]); f(x)=y <-> y:=x*(x-r.g);

preprocess() <-> (solverInput.g(v)) -> (r.g:=v) otherwise (r.g:=4);

initialstate()=x0 <-> x0:=0, e.g:=0.01;

finalstate(x) <-> f(x)<f(x-e.g), f(x)<f(x+e.g);

genA(x)=Es <-> Es:=[-e.g, e.g];

delta(x,e)=x2 <-> x2:=x+e, constraint(x2);

constraint(x) <-> x >= 0, x<=20;

goal(x) = f(x);

st(x) <-> finalstate(x), solverOutput.g:=x;
付録2　入力出力のグラフ表示

//opt13.set

inputsequence0()=Rs <-> Rs:=[1,2,3,4,5,6];

initialstate0()=c <-> c:=0, hist.g:=[];

delta0(c,r)=cc <-> 


cc:=call_solver("opt11.p",r),


showHistory(r,cc);

showHistory(r,cc) <->


hist1:=hist.g,


hist2:=append(hist1,[[r,cc]]),


show2(2,transpose(hist2),"trajectory"),


hist.g:=hist2;

付録3　2変数の値を取得するSolver

//opt21.set
func([f]); f(x)=y <-> y:=x*(x-r.g);

preprocess() <-> 

   (solverInput.g([v1,v2])) -> (x0.g:=v1,r.g:=v2) otherwise (x0.g:=0, r.g:=4);

initialstate()=x0 <-> x0:=x0.g, e.g:=0.01;

finalstate(x) <-> f(x)<f(x-e.g), f(x)<f(x+e.g);

genA(x)=Es <-> Es:=[-e.g, e.g];

delta(x,e)=x2 <-> x2:=x+e, constraint(x2);

constraint(x) <-> x >= 0, x<=20;

goal(x) = f(x);

st(x) <-> finalstate(x), solverOutput.g:=x;
付録4　Solverを状態遷移関数とするオートマトン

//opt22.set 

inputsequence0()=Rs <-> Rs:=[1,2,3,4,5,6];

initialstate0()=c <-> c:=0;

delta0(c,r)=cc <-> cc:=call_solver("opt21.p", [c,r]);

